Paper SAS1755-2015

Working with Panel Data: Extracting Value from Multiple Customer Observations

Roberto G. Gutierrez and Kenneth Sanford, SAS Institute Inc.

ABSTRACT

Many retail and consumer packaged goods (CPG) companies are now keeping track of what their customers purchased in the past, often through some form of loyalty program. This record keeping is one example of how modern corporations are building data sets that have a panel structure, a data structure that is also pervasive in insurance and finance organizations. Panel data (sometimes called longitudinal data) can be thought of as the joining of cross-sectional and time series data. Panel data enable analysts to control for factors that cannot be considered by simple cross-sectional regression models that ignore the time dimension. These factors, which are unobserved by the modeler, might bias regression coefficients if they are ignored. This paper compares several methods of working with panel data in the PANEL procedure and discusses how you might benefit from using multiple observations for each customer. Sample code is available.

INTRODUCTION

Panel data occur when a panel of individuals—people, households, corporations, or otherwise—are observed over a period of time during which several observations per individual are obtained. Panel data have two dimensions: the individual dimension (or *cross section*) and the time dimension. As an example, suppose you follow a panel of households for one year. Each month you record the details of their purchases at a certain grocery chain, along with other demographic factors, such as household size, favorite grocery store, and whether the household receives government assistance. Because you obtain data every month, you have multiple records per household. The biggest advantage of the panel structure is that the multiple observations afford you more flexibility in how you can model the purchases of these households. You can determine the effect of, say, a new promotional campaign because any household characteristics you might have neglected to measure do not confound the analysis as they would if you had only one observation per household. Put simply, the multiple observations per household act as their own control group. Sometimes the household takes advantage of the new promotion, and sometimes it doesn't, but the household otherwise remains the same.

Analyzing panel data is fundamental to econometrics. The texts from Baltagi (2008) and Wooldridge (2010) are among the most complete treatments of the topic.

Formally, for a panel of N individuals, consider the linear regression model

$$y_{it} = \beta_0 + \boldsymbol{\beta}_x \mathbf{X}_{it} + \boldsymbol{\beta}_z \mathbf{Z}_i + \nu_i + \epsilon_{it}$$

where i denotes the individual and t is any one of T time points. The regression model has two sets of explanatory variables: a set of **X** variables that vary over time (time-varying, or TV) and a set of **Z** variables that do not vary over time (time-invariant, or TI). The ν_i are known as individual (or cross-sectional) effects, and the ϵ_{it} are the observation-level regression errors.

There are several ways to fit the preceding regression model, and each strategy differs in what it is willing to assume about the explanatory variables, the individual effects, the observation-level errors, and their relationships.

You fit linear regression models to panel data by using the PANEL procedure in SAS/ETS® software. This paper demonstrates several available methods, gives details about each method's assumptions, and interprets the results. Some of these methods and features are new in SAS/ETS 14.1.

The next section describes the data set used throughout this paper. The subsequent sections cover four different estimation strategies, as applied to these data.

GROCERY DATA

The following consumer-loyalty data are from 330 households who shopped regularly at a grocery chain in the Raleigh, North Carolina, area. The data track monthly meat expenditures for the year 2011. There are 12 monthly observations per household; some observations are missing because the household did not visit the chain during that month (or did not use its loyalty card).

The following SAS statements create the data set Grocery:

```
data Grocery;
   input Houseid Month Meat Govt Hsize Rural Alcohol MealsOut;
   label Meat     = 'Meat purchases per store visit';
   label Govt     = '1 if used government assistance that month';
   label Hsize    = 'Household size';
   label Rural    = '1 if rural location visited at least once';
   label Alcohol  = '1 if at least 10% spent on alcohol';
   label MealsOut = 'Meals per week outside of household (survey)';
datalines;
1   1  55.841  1  5  0  1  3
1   3  49.372  1  5  0  1  3
1   4   59.43  1  5  0  1  3
1   5   52.25  1  5  0  1  3
1   6  41.623  1  5  0  0  3
1   7  59.357  1  5  0  1  3
1   9  58.512  1  5  0  0  3
1  10   46.15  1  5  0  0  3
1  11  47.027  1  5  0  0  3
1  12  56.065  1  5  0  0  3
2   1  19.949  0  4  1  0  6
2   2  15.327  0  4  1  1  6
2   3  27.836  0  4  1  0  6
2   4  22.943  0  4  1  0  6

   ... more lines ...
```

The variables **HouseID** and **Month** are identification variables that represent the household and month, respectively. The dependent variable **Meat** records the average amount per visit spent on butcher meats. The variable **Govt** has a value of 1 if government assistance (such food stamps and WIC) was used during that month; **Hsize** is the household size; **Rural** records whether a rural store location was visited; and **Alcohol** has a value of 1 if at least 10% of the household's expenditures for the month were for alcoholic beverages. The variable **MealsOut** records the number of meals per week outside the household, as provided on a survey the household filled out when it applied for its loyalty card.

As always with panel data, it is vital that you keep track of which variables in your data vary within households (the TV variables) and which are constant (the TI variables). In the data used for this paper, the variables **Hsize** and **MealsOut** are TI, and the remaining variables are TV because each varies within at least one household.

You want to determine the association between government assistance and meat purchases while controlling for other available factors. Therefore, throughout this paper you consider variations of the panel linear regression model

$$\textbf{Meat}_{it} = \beta_0 + \beta_1\textbf{Govt}_{it} + \beta_2\textbf{Hsize}_i + \beta_3\textbf{Rural}_{it} + \beta_4\textbf{Alcohol}_{it} + \beta_5\textbf{MealsOut}_i + \nu_i + \epsilon_{it}$$

for household i during month t. The only differences among the model variations are the assumptions about the relationship between the household effect ν_i and the explanatory variables.

RANDOM-EFFECTS ESTIMATION

You begin with the random-effects model because it is commonly found in the literature, regardless of field of study. This model is referred to as a random-effects model because the error terms ν_i and ϵ_{it} are randomly (and independently) drawn from some large population. That is, their values are determined without regard to what else is in the model.

Because the individual effects (ν_i) are random, you can treat them as nuisance parameters. You estimate their variance, use the variance to sweep out the individual effects, and then apply standard least squares techniques to estimate the regression coefficients. This process is known as generalized least squares (GLS), and you often see the terms GLS and random effects used interchangeably.

Random-effects estimation provides consistent and precise estimates of the regression coefficients, provided that the individual effects are truly random and uncorrelated with the explanatory variables.

Using the PANEL Procedure

The following statements fit a random-effects model to the grocery data:

```
proc panel data = Grocery;
   id HouseID Month;
   model Meat = Govt Hsize Rural Alcohol MealsOut / ranone;
run;
```

In the preceding code, the ID statement specifies the panel and time variables, in that order. The MODEL statement specifies the dependent and explanatory variables, and the RANONE option requests random-effects estimation.

Figure 1 Random-Effects Estimation

The PANEL Procedure
Wansbeek and Kapteyn Variance Components (RanOne)

Dependent Variable: Meat (Meat purchases per store visit)

Model Description	
Estimation Method	RanOne
Number of Cross Sections	330
Time Series Length	12

Fit Statistics			
SSE	84930.9948	**DFE**	3567
MSE	23.8102	**Root MSE**	4.8796
R-Square	0.1232		

Variance Component Estimates	
Variance Component for Cross Sections	190.123
Variance Component for Error	24.99832

Hausman Test for Random Effects

Coefficients	DF	m Value	Pr > m
3	3	25.72	<.0001

Parameter Estimates

Variable	DF	Estimate	Standard Error	t Value	Pr > |t|	Label
Intercept	1	20.50606	2.3327	8.79	<.0001	Intercept
Govt	1	5.050562	0.5989	8.43	<.0001	1 if used government assistance that month
Hsize	1	5.145648	0.4774	10.78	<.0001	Household size
Rural	1	-1.41068	0.3449	-4.09	<.0001	1 if rural location visited at least once
Alcohol	1	2.982397	0.1960	15.22	<.0001	1 if at least 10% spent on alcohol
MealsOut	1	-2.82761	0.3848	-7.35	<.0001	Meals per week outside of household (survey)

Figure 1 provides the estimation results. Here is the guided tour:

- The “Model Description” table simply verifies the estimation method, the number of households, and the (maximum) number of time points per household.

- The “Fit Statistics” table provides summary statistics such as R^2, analogous to what you would find in standard linear regression.

- The “Variance Component Estimates” table lists the estimated variances for both components: the variance of the individual effects $\hat{\sigma}_\nu^2 = 190.1$, and the variance of the overall errors $\hat{\sigma}_\epsilon^2 = 25.0$.

  Because you didn’t specify otherwise, PROC PANEL used the Wansbeek and Kapteyn (1989) method to estimate the variance components, as noted in the title of the output. Three alternative methods are also available and can be specified through the VCOMP= option in the MODEL statement; for more information, see the chapter about the PANEL procedure in the *SAS/ETS 14.1 User’s Guide*. In practice, it makes little difference which method you use with moderate to large data sets.

- The “Hausman Test for Random Effects” table provides the model specification test, as described by Hausman (1978). Think of this test as a referendum on the random-effects strategy. The null hypothesis is that the random-effects model is, in fact, appropriate for your data.

  That the null hypothesis is soundly rejected is a problem that casts doubt on the validity of the random-effects estimator. In the following sections, you will find some alternative strategies.

- The “Parameter Estimates” table is a standard table of regression coefficients. If you believed this model, you would conclude that households on government assistance purchase about $5.05 more in meat products per visit, controlling for other factors such as household size and rural store location. However, the results of the Hausman test invalidate that conclusion.

Correlated Individual Effects

The results of the Hausman test tell you that you need to consider an alternative estimator. Before proceeding, it is worthwhile to look more closely at this issue.

Formally, the random-effects strategy assumes the following:

A. ν_i and ϵ_{it} follow a normal (or similar) distribution.

B. ν_i and ϵ_{it} are uncorrelated with each other.

C. ϵ_{it} is uncorrelated with each explanatory variable.

D. ν_i is uncorrelated with each explanatory variable.

An explanatory variable is known as *exogenous* if it satisfies C and D, and *endogenous* if it violates one or both.

The Hausman test is a test of assumption D, and thus the problem with the random-effects strategy is that household-level effects are correlated with one or more explanatory variables. Think of the household effects as latent propensities to spend. Perhaps being on government assistance is associated with a tendency to spend more (or less) on meat products, in a way not adequately explained by a single regression coefficient of a yes/no variable. Perhaps it was not enough to determine which households were on government assistance—it would have been better to record the precise amount of assistance that each household received. In other words, you can think of endogeneity as a form of measurement error.

Viewing endogeneity from a different angle, consider the regression coefficient of the variable **Govt** from the “Parameter Estimates” table in Figure 1. You can interpret this coefficient in two ways:

- as the effect on a particular household that enrolled in government assistance at some point in 2011

- as the difference between two households, one that was always on government assistance and one that never was

The first interpretation is known as the *within effect*, the second as the *between effect*. If you confine your estimation strategy to use only within-household data, you obtain the *within estimator* that directly estimates the within effect. Likewise, if you confine yourself to comparing only household averages, you obtain the *between estimator*.

When you assume that individual effects are uncorrelated, you assume that the between effect and the within effect are one and the same. Your GLS estimator pools the within-household and between-households data to form a more efficient estimator. Because you use two sources of information, you gain more precision.

Figure 2 shows hypothetically what can happen when you have correlated individual effects. The graph depicts a situation where the **x** variable increases over 10 groups and is thus correlated with the group effects. The slope within groups hovers around –1, whereas the slope between groups is +1. The GLS slope as obtained from the random-effects estimator is a weighted average of the within and between slopes. When you have correlated individual effects, the GLS slope is useless; it represents neither the within effects nor the between effect.

Figure 2 Correlated Individual Effects



When your between and within estimators differ, you should always favor the within estimator. Regardless of correlation, the within estimator is consistent for the true regression coefficient and more indicative of a causal effect. Because all household characteristics remain unchanged while you are comparing data within that household, there is no possible way that an unobserved household effect can confound the within estimator.

In the hypothetical graph in Figure 2, the within slope of –1 is the one you are after. When individual effects are correlated, the GLS slope is merely a biased version of the within slope.

To summarize, if your Hausman test rejects the null hypothesis of uncorrelated individual effects, you should reject the GLS estimator in favor of the within estimator. In the process, you avoid using any between-individuals data that serve only to bias your results. If you fail to reject the null hypothesis of the Hausman test, then you should favor the GLS estimator for its added efficiency.

FIXED-EFFECTS ESTIMATION

The previous section indicates that you should use the within estimator for this model and these data. The within estimator treats the individual (household) effects as fixed and is thus also called the *fixed-effects estimator*. Because the effects are fixed, you can treat them as regression coefficients in a standard regression model, to be estimated along with the coefficients of **Govt**, **Hsize**, and so on.

You obtain fixed-effects estimates by using the FIXONE option in the MODEL statement:

```
proc panel data = Grocery;
   id HouseID Month;
   model Meat = Govt Hsize Rural Alcohol MealsOut / fixone;
run;
```

Figure 3 shows the results. By default, estimates of the household effects are not provided, but you can get them by specifying the PRINTFIXED option in the MODEL statement. Most of the time it suffices to know whether the household effects, when taken as a whole, are significant to the analysis. The "F Test for No Fixed Effects" table helps answer that question. The null hypothesis is that all household effects equal 0, and you reject that hypothesis.

Figure 3 Fixed-Effects Estimation

The PANEL Procedure
Fixed One-Way Estimates

Dependent Variable: Meat (Meat purchases per store visit)

Model Description	
Estimation Method	FixOne
Number of Cross Sections	330
Time Series Length	12

Fit Statistics			
SSE	80994.5460	**DFE**	3240
MSE	24.9983	**Root MSE**	4.9998
R-Square	0.9055		

F Test for No Fixed Effects			
Num DF	**Den DF**	**F Value**	**Pr > F**
329	3240	32.06	<.0001

Parameter Estimates

Variable	DF	Estimate	Standard Error	t Value	Pr > |t|	Label
Intercept	1	53.89442	1.6500	32.66	<.0001	Intercept
Govt	1	3.591205	0.6650	5.40	<.0001	1 if used government assistance that month
Hsize	0	0	.	.	.	Household size
Rural	1	-1.45444	0.3578	-4.07	<.0001	1 if rural location visited at least once
Alcohol	1	2.992035	0.2013	14.86	<.0001	1 if at least 10% spent on alcohol
MealsOut	0	0	.	.	.	Meals per week outside of household (survey)

The coefficient of the variable **Govt** is now 3.59, which you take to be a more accurate representation of the effect of government assistance on meat purchases. The standard error of this estimate is larger than it was with random effects (0.67 versus 0.60), but that is a small price to pay for consistency.

An unfortunate side effect of fixed-effects estimation is that the variables **Hsize** and **MealsOut** were dropped from the model; they were dropped because they are constant within households, and thus you cannot estimate their effects by using only within-household data. Instead, you might consider an estimator that is both consistent and able to estimate effects for time-invariant (TI) variables. One such estimator is described later in this paper.

BETWEEN-EFFECTS ESTIMATION

In the section on random-effects estimation, the Hausman test indicates that the within and between estimators are different. In the section on fixed-effects estimation, you obtain the consistent within estimator. To obtain the between estimator, use the BTWNG option in the MODEL statement:

```
proc panel data = Grocery;
   id HouseID Month;
   model Meat = Govt Hsize Rural Alcohol MealsOut / btwng;
run;
```

Figure 4 shows the results. The estimation is equivalent to performing linear regression on the household-level means for all variables.

Figure 4 Hausman-Taylor Estimation

The PANEL Procedure
Between-Groups Estimates

Dependent Variable: Meat (Meat purchases per store visit)

Model Description	
Estimation Method	BtwGrps
Number of Cross Sections	330
Time Series Length	12

Parameter Estimates

Variable	DF	Estimate	Standard Error	t Value	Pr > |t|	Label
Intercept	1	16.98442	1.7004	9.99	<.0001	Intercept
Govt	1	13.40059	0.9886	13.56	<.0001	1 if used government assistance that month
Hsize	1	5.092447	0.3032	16.80	<.0001	Household size
Rural	1	0.005439	1.4038	0.00	0.9969	1 if rural location visited at least once
Alcohol	1	1.082457	1.7681	0.61	0.5408	1 if at least 10% spent on alcohol
MealsOut	1	-2.67669	0.2629	-10.18	<.0001	Meals per week outside of household (survey)

Because you lose much information by collapsing the data into averages, you should not rely on the between estimator for anything other than to help diagnose correlated individual effects. By comparing the between estimates to the within estimates, you can determine where the bias in GLS occurs. Of course, you can detect bias in GLS by directly comparing it to the within estimator, but using the between estimator makes the bias more obvious.

To illustrate, the GLS estimate of the coefficient of **Govt** is 5.05, the within estimate is 3.59, and the between estimate is 13.40. The bias is much more evident when you compare 3.59 to 13.40, knowing that under GLS these should estimate the same quantity!

HAUSMAN-TAYLOR ESTIMATION

The problem of choosing between random effects and fixed effects is fairly standard in econometrics. You let the Hausman test tell you which way to go: reject the null hypothesis, use fixed effects; fail to reject, use the more efficient random effects.

Consider, then, the case where a Hausman test indicates the fixed-effects estimator but you are not satisfied with that. In particular, you don't like that you cannot estimate coefficients for time-invariant variables. One solution is the Hausman-Taylor estimator.

Hausman and Taylor (1981) describe an instrumental-variables approach to dealing with endogeneity due to correlated individual effects. You specify which variables you think are correlated with the individual effects, and the estimator derives a set of instruments based on the uncorrelated variables, their individual-level averages, and the deviations of the variables from these averages. The PANEL procedure fits the model by two-stage least squares (2SLS). During

the first stage, the instruments are used to "predict" the correlated variables. At the second stage, estimation proceeds with a modified GLS strategy (for more information, see Baltagi 2008, sec. 7.4).

The Hausman-Taylor estimator is new to the PANEL procedure in SAS/ETS 14.1.

In light of our previous comparison of the coefficients of the variable **Govt**, that variable is a prime candidate for being correlated with the household effects. Also, based on experience, you believe that the variable **MealsOut** might be correlated with the household effects. For your grocery data, it makes sense that the more often a family eats out, the less often it buys meat from the grocery store. It could be that this correlation is not adequately described by a single regression term.

You specify **Govt** and **MealsOut** as correlated by specifying an INSTRUMENTS statement immediately before your MODEL statement, as follows:

```
proc panel data = Grocery;
   id HouseID Month;
   instruments correlated = (Govt MealsOut);
   model Meat = Govt Hsize Rural Alcohol MealsOut / htaylor;
run;
```

Figure 5 provides the results of Hausman-Taylor estimation. Note the following:

- The "Variance Component Estimates" table provides variance estimates for both the household effects (cross sections) and the overall errors, similar to what you get with random effects. If you compare these values to the corresponding ones in Figure 1, you'll see that the variance of household effects is now much different (97.3 versus 190.1). Because these are nuisance parameters, you should not read too much into that difference except to say that the random-effects estimate is likely to be biased, given what you know about the household effects and their correlation with the explanatory variables. If your theory that **Govt** and **MealsOut** are the culprits holds true, then you should favor the Hausman-Taylor variance.
- The "Hausman Test against Fixed Effects" table provides a Hausman test that is similar to the one you get with random effects. The Hausman test compares the Hausman-Taylor estimator to the within estimator. Think of this test as a referendum on your choice of correlated variables. The null hypothesis is that you made an adequate choice, and that hypothesis seems to hold, given the results.
- The "Parameter Estimates" table is similar to those in previous sections, with added columns that mark the variables that are assumed to be correlated (C) and the variables that are time-invariant (TI).
- The coefficients for the time-varying variables (**Govt**, **Rural**, and **Alcohol**) are now more in agreement with those from the fixed-effects estimation in Figure 3. That is, they appear to be consistent, as also evidenced by the Hausman test.
- By stipulating the correlated variables and using the Hausman-Taylor model, you also obtain coefficients for the time-invariant variables, which was not possible in fixed-effects estimation.

Figure 5 Hausman-Taylor Estimation

The PANEL Procedure
Hausman and Taylor Model for Correlated Individual Effects (HTaylor)

Dependent Variable: Meat (Meat purchases per store visit)

Model Description	
Estimation Method	HTaylor
Number of Cross Sections	330
Time Series Length	12

Fit Statistics			
SSE	89144.7983	**DFE**	3567
MSE	24.9915	**Root MSE**	4.9992
R-Square	0.1547		

Figure 5 *continued*

Variance Component Estimates	
Variance Component for Cross Sections	97.29627
Variance Component for Error	24.97519

Hausman Test against Fixed Effects			
Coefficients	**DF**	**m Value**	**Pr > m**
3	1	0.76	0.3824

Parameter Estimates

Variable	Type		DF	Estimate	Standard Error	t Value	Pr > |t|	Label
Intercept			1	19.12589	2.4038	7.96	<.0001	Intercept
Govt	C		1	3.583391	0.6649	5.39	<.0001	1 if used government assistance that month
Hsize		TI	1	5.17389	0.3523	14.68	<.0001	Household size
Rural			1	-1.43991	0.3573	-4.03	<.0001	1 if rural location visited at least once
Alcohol			1	2.974996	0.2004	14.85	<.0001	1 if at least 10% spent on alcohol
MealsOut	C	TI	1	-1.92242	0.8090	-2.38	0.0175	Meals per week outside of household (survey)

C: correlated with the individual effects
TI: constant (time-invariant) within cross sections

If you go back and compare the between estimators to the within estimators (for all coefficients, not just that of **Govt**), you find disagreement for all three time-varying variables. The GLS estimator is biased for all three. You were able to fix the bias in all three variables by specifying **Govt** (and the TI variable **MealsOut**) as correlated. That was fortunate, but not something you should expect. In general, stipulating a variable as correlated will alleviate the bias for that variable, but there is no guarantee that it will fix the bias in the other variables. If it doesn't, you can specify more correlated variables. However, there is a limit. You must have at least one uncorrelated TV variable for every correlated TI variable. If that seems complicated, don't worry; PROC PANEL warns you if you take things too far. Your goal is to eliminate systematic bias, and you can use the Hausman test as a guide.

Finally, you should realize that the Hausman-Taylor estimator is not a cure-all for correlated individual effects. Your data need to be able to predict the correlated variables. Otherwise you run into what is known in econometrics as the problem of weak instruments. If you have weak instruments, you will obtain biased estimates with very large standard errors. In the Hausman-Taylor output, the standard error for **MealsOut** is somewhat large, but not large enough to be of any real concern.

OTHER METHODS

The estimators that are considered here represent only a fraction of what you can do with the PANEL procedure. In addition to the methods demonstrated here, you can do the following:

- Fit two-way fixed-effects and random-effects models. In addition to cross-sectional effects, these models have effects that are specific to time periods across cross sections.
- Perform estimation for dynamic panel models, models that include lagged versions of the dependent variable as explanatory variables.
- Fit panel models that adjust for serial correlation, heteroscedasticity, and clustering.
- Perform a series of unit-root tests that determine whether the dependent variable is stationary over time.
- Perform model specification tests such as the Durbin-Watson (1951) test for serial correlation. The PANEL procedure supports over a dozen specification tests.
- Obtain the Amemiya and MaCurdy (1986) estimator, which is closely related to the Hausman-Taylor estimator.

For more information, see the chapter about the PANEL procedure in the *SAS/ETS 14.1 User's Guide*.

SUMMARY

Panel data provide an internal control structure that enables you to fit regression models that are free from the confounding caused by unobserved individual effects. You can choose from many estimation strategies, depending on the properties of your explanatory variables. Hausman tests and prior experience can guide you to choose the most appropriate strategy for your data. You fit these models by using the PANEL procedure. The Hausman-Taylor estimator is a new feature of PROC PANEL in SAS/ETS 14.1.

REFERENCES

Amemiya, T., and MaCurdy, T. E. (1986). "Instrumental–Variable Estimation of an Error Components Model." *Econometrica* 54:869–881.

Baltagi, B. H. (2008). *Econometric Analysis of Panel Data*. 4th ed. New York: John Wiley & Sons.

Durbin, J., and Watson, G. S. (1951). "Testing for Serial Correlation in Least Squares Regression." *Biometrika* 37:409–428.

Hausman, J. A. (1978). "Specification Tests in Econometrics." *Econometrica* 46:1251–1271.

Hausman, J. A., and Taylor, W. E. (1981). "Panel Data and Unobservable Individual Effects." *Econometrica* 49:1377–1398.

Wansbeek, T., and Kapteyn, A. (1989). "Estimation of the Error-Components Model with Incomplete Panels." *Journal of Econometrics* 41:341–361.

Wooldridge, J. M. (2010). *Econometric Analysis of Cross Section and Panel Data*. 2nd ed. Cambridge, MA: MIT Press.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors:

Roberto G. Gutierrez
SAS Institute Inc.
SAS Campus Drive
Cary, NC 27513
Bobby.Gutierrez@sas.com

Kenneth Sanford
SAS Institute Inc.
SAS Campus Drive
Cary, NC 27513
Kenneth.Sanford@sas.com